



Cálculo Diferencial e Integral: um KIT de sobrevivência

Prof. Doherty Andrade

Campos Conservativos

Vamos ilustrar o uso do Maple no cálculo de potenciais de campos conservativos. Primeiramente relembremos algumas definições e resultados básicos.

1. Introdução

Se $F: \mathbb{R}^3 \rightarrow \mathbb{R}^3$ está definido sobre a região Ω do $\mathbb{R}^3$ e existe $g: \mathbb{R}^3 \rightarrow \mathbb{R}$ tal que $\text{grad}(g) = F$, dizemos que F é um campo conservativo e que g é um potencial para F .

No cálculo de gradientes, vamos usar um pacote de Álgebra Linear. Vamos lembrar a definição de gradiente.

> **with(linalg):**

Warning, new definition for norm

Warning, new definition for trace

> **'grad(g,[x,y,z])'=[Diff(g,x),Diff(g,y),Diff(g,z)];**

$$\text{grad}(g, [x, y, z]) = \left[\frac{\partial}{\partial x} g, \frac{\partial}{\partial y} g, \frac{\partial}{\partial z} g \right]$$

Exemplo 1: Seja $F(x, y, z) = [y z, x z, x y]$. Este é um campo conservativo e $g(x, y, z) = x y z$ é um potencial para F .

> **grad(x*y*z, [x,y,z]);**

$$[y z, x z, x y]$$

Lembramos que se $\text{grad}(g) = F$, e se $r(t) : R \rightarrow R^3$ é uma função diferenciável definida sobre o intervalo $[a, b]$, então

$\frac{\partial}{\partial t} g(r(t)) = \text{grad}(g)(r(t)) \cdot \frac{\partial}{\partial t} r(t) = F(r(t)) \cdot r'(t)$. Isto é Regra da Cadeia para derivadas de funções compostas.

Assim, se integrarmos de $t=a$ a $t=b$,

$$\int_a^b F(r(t)) \cdot r'(t) dt = g(r(b)) - g(r(a)).$$

Concluimos que se F é um campo conservativo e $r(t)$ descreve uma curva fechada C , então $\int_C F \cdot dr = 0$, onde a integral é tomada sobre a curva fechada.

Exemplo 2: A função $F(x,y,z) = [y, -x, 0]$ não é um campo conservativo, pois tomando a curva como sendo um círculo de raio 1 no plano x-y, vemos que

> **Int(dotprod([sin(t),-cos(t)],[diff(cos(t),t),diff(sin(t),t)]),t=0..2*Pi)**
= int(dotprod([sin(t),-cos(t)],[diff(cos(t),t),diff(sin(t),t)]),t=0..2*Pi);

$$\int_0^{2\pi} -\sin(t)^2 - \cos(t)^2 dt = -2\pi$$

2. Como reconhecer se F é conservativo?

Teorema 1 : Seja F contínuo em uma região Ω do $\mathbb{R}^3$. São equivalentes:

(a) F é conservativo.

(b) A integral de linha $\int F(x) dx$ é independente do caminho.

Este é um ótimo resultado. A parte mais interessante deste teorema é a implicação (b) implica (a).

Exemplo 1 (de novo): Vimos que $F(x, y, z) = [y z, x z, x y]$ é conservativo, $\mathbb{R}^3$

assim pelo Teorema 1, a integral sobre qualquer caminho fechado do zero. Tomemos como exemplo o círculo no plano x-y.

```
> F:=(x,y,z)->[y*z,x*z,x*y];
x:=t->cos(t); y:=t->sin(t); z:=t->0;
Int('F(x(t),y(t),z(t))[1]''D(x)(t)',t=0..2*Pi)
+ Int('F(x(t),y(t),z(t))[2]''D(y)(t)',t=0..2*Pi)
+ Int('F(x(t),y(t),z(t))[3]''D(z)(t)',t=0..2*Pi) =
int(F(x(t),y(t),z(t))[1]*D(x)(t),t=0..2*Pi)
+ int(F(x(t),y(t),z(t))[2]*D(y)(t),t=0..2*Pi)
+ int(F(x(t),y(t),z(t))[3]*D(z)(t),t=0..2*Pi);
```

$$F := (x, y, z) \rightarrow [y z, x z, x y]$$

$$x := \cos$$

$$y := \sin$$

$$z := 0$$

$$\int_0^{2\pi} F(x(t), y(t), z(t))_1 D(x)(t) dt + \int_0^{2\pi} F(x(t), y(t), z(t))_2 D(y)(t) dt + \int_0^{2\pi} F(x(t), y(t), z(t))_3 D(z)(t) dt = 0$$

Exemplo 2 (de novo): Vimos que a integral sobre qualquer curva fechada não é necessariamente zero para este exemplo. Vamos ver, de outro modo, que não existe g tal que

$$\frac{\partial}{\partial x} g = y \quad \text{and} \quad \frac{\partial}{\partial y} g = -x.$$

Se existisse tal g , então as derivadas parciais mistas deveriam ser iguais.

Isto é,

$$\frac{\partial^2}{\partial x \partial y} g = \frac{\partial^2}{\partial y \partial x} g$$

$$-1 = 1.$$

Uma contradição.

3. Como reconhecer se F é conservativo?

Teorema 2: Seja $F(x, y, z) = [P(x, y, z), Q(x, y, z), R(x, y, z)]$ e suponha que P , Q , e R , juntos com suas primeiras derivadas são contínuas em uma região simples Ω . São equivalentes:

(a) F é um campo conservativo sobre Ω ,

(b) $\frac{\partial}{\partial y} P = \frac{\partial}{\partial x} Q$, $\frac{\partial}{\partial z} P = \frac{\partial}{\partial x} R$, e $\frac{\partial}{\partial z} Q = \frac{\partial}{\partial y} R$ em Ω .

Exemplo 1: Vamos verificar as derivadas mistas com o campo $F(x, y, z) = [yz, xz, xy]$.

```
> x:='x': y:='y': z:='z':
F:=(x,y,z)->[y*z,x*z,x*y];
diff(F(x,y,z)[1],y),diff(F(x,y,z)[2],x);
diff(F(x,y,z)[1],z),diff(F(x,y,z)[3],x);
diff(F(x,y,z)[3],y),diff(F(x,y,z)[2],z);
```

$$F := (x, y, z) \rightarrow [yz, xz, xy]$$

z, z

y, y

x, x

Exemplo 2: o campo não é conservativo

```
> F:=(x,y,z)->[y,-x,0];  
diff(F(x,y,z)[1],y),diff(F(x,y,z)[2],x);  
diff(F(x,y,z)[1],z),diff(F(x,y,z)[3],x);  
diff(F(x,y,z)[3],y),diff(F(x,y,z)[2],z);
```

$F := (x, y, z) \rightarrow [y, -x, 0]$

$1, -1$

$0, 0$

$0, 0$

4. Como encontrar o potencial?

É importante saber como determinar o potencial para um campo conservativo. A técnica é simples e poder ser programada.

O Maple V tem uma rotina para determinar o potencial de um campo. Veja os comandos.

Exemplo 1 (mais uma vez):

```
> F:=(x,y,z)->[y*z,x*z,x*y];  
potential(F(x,y,z),[x,y,z],'h');  
h;
```

$F := (x, y, z) \rightarrow [y z, x z, x y]$

$true$

$x y z$

Exemplo 2 (mais uma vez):

```
> F:=(x,y,z)->[y,-x,0];  
potential(F(x,y,z),[x,y,z],'g');
```

$$F := (x, y, z) \rightarrow [y, -x, 0]$$

false

Exemplo 3:

```
> F:=(x,y,z)->[exp(y+2*z),x*exp(y+2*z),2*x*exp(y+2*z)];
potential(F(x,y,z),[x,y,z],'g');
g;
```

$$F := (x, y, z) \rightarrow [e^{(y+2z)}, x e^{(y+2z)}, 2 x e^{(y+2z)}]$$

true

$$x e^{(y+2z)}$$

Exemplo 4:

```
> F:=(x,y,z)->[exp(x)*sin(y)+2*y,exp(x)*cos(y)+2*x-2*y,0];
potential(F(x,y),[x,y,z],'g');
g;
grad(exp(x)*sin(y)+2*x*y-y^2,[x,y,z]);
```

$$F := (x, y, z) \rightarrow [e^x \sin(y) + 2y, e^x \cos(y) + 2x - 2y, 0]$$

true

$$2yx + e^x \sin(y) - y^2$$

$$[e^x \sin(y) + 2y, e^x \cos(y) + 2x - 2y, 0]$$